New Features for Pulse Programs and Xwin-NMR 3.5

Detlef Moskau

March 2005

New Features for Pulse Programs

Frequency Lists in Pulse Programs

Define frequencies via constants

- **XWIN-NMR 3.5 allows to set frequencies within a pulse program without the need of external frequency lists.**
- **Example:**

```
.....
3 d11 fq=cnst23(bf ppm):f2
....
fq:         take BF2 (as :f2 is used),
            add a frequency cnst23*BF2*10e-6

SFO2= BF2 + cnst23*BF2*10e-6

cnst23:   could be C-alpha chemical shift, e.g.
```

Calculate Excitation Offset for Shaped Pulses within a Pulse Programs

spoff calculation

- Excitation offsets for shaped pulses can be defined directly in a pulse program:

example 1:

```
;hbhaconhgp3d
;avance version (02/05/31)
;HBHACONH

"spoffs2=0"
"spoffs3=0"
"spoffs5=bf2*((cnst21-cnst23)/1000000)"

;cnst21: CO chemical shift (offset, in ppm)
;cnst23: Caliphatic chemical shift (offset, in ppm)
```

spoff calculation

- **Excitation offsets for shaped pulses can be defined directly in a pulse program:**

example 2:

```
;hbhanhgp3d
;avance version (02/05/31)
;HBHANH

"spoffs2=0"
"spoffs3=0"
"spoffs5=bf2*(cnst21/1000000)-o2"

;cnst21: CO chemical shift (offset, in ppm)
```

Alignment of Pulses

Pulse Alignment

- XWIN-NMR 3.5 allows alignment of pulses in a very easy way
- Example:

(center (p2 ph1):f1 (p14:sp3 ph1):f2) center pulses

(ralign (p1 ph2) (p3 ph9):f2) right alignment of shorter pulse

(lalign (p1 ph2) (p3 ph9):f2) left alignment of shorter pulse

(reference (p1 ph2) center (p3 ph9):f2 ralign (p21 ph2):f3)
use p1 as reference for alignment

Pulse Alignment

- Simultaneous pulses do not need to be coded on the same line
- Example:

(reference (p1 ph2) center (p3 ph9):f2 ralign (p21 ph2):f3)

OR:

```
(
        reference (p1 ph2)
        center (p3 ph9):f2
        ralign (p21 ph2):f3
)
```

Pulse Alignment

```
(
   reference (d0 p1 ph2 d0) :f1
   center (p2 ph3) :f2
   ralign (p3 ph3) :f3
)
```



p1 > p2
p1 > p3



p1 < p2
p1 > p3

Pulse Alignment

- New syntax for delays
- Example:

"DELTA=d0*2+larger(p14,p22)-p14+36u“

New syntax for pulsed field gradients

Pulse Programming: PFG's

XWINNMR3.1 options for programming gradients

GRADIENT(cnst21)	gradient of length *p16* '(GS-syntax') gradient sequence and assignment of gradient strength for individual pulses are defined in *GRDPROG*
2u:ngrad	*set next value according to GRDPROG*
gron1, groff	*gradient ON / OFF according to GPZ1*
p16:gp1	gradient of length p16, gradient strength = *gpz1 (gpx1, gpy1)* gradient shape defined by *gpnam1*
*p16:gp1*EA*	multiplier for gradient strength; allow to invert gradient sign (useful for echo-antiecho gradient selection) only XWIN-NMR version >= 2.5
2grad EA	invert gradient multiplier EA

Pulse Programming: PFG's

- Syntax *GRADIENT(cnst21)* not used anymore
- Syntax *ngrad* not used anymore
- Pulse programs **gs** substituted by **gp**
- Content of *GRDPROG* is now coded within pulse program

Pulse Programming: PFG's

```
;imgegp3d
.....
lgrad r2d<2d> = td1
lgrad r3d<3d> = td2
......

 5u                    grad{ (0)+r3d(5.057) | (0)+r2d(5.057) | (3.112)  }
 d27
 5u                    grad{ (0)          | (0)          | (0)      }
 d15
 5u                    grad{ (0)          | (0)          | (-6.739) }
 d21
 ACQ_START(ph30,ph31)
 aq DWELL_GEN:f1
 5u                    grad{ (0)          | (0)          | (0)      }
 rcyc=2
 5m st igrad r2d
 lo to 2 times td1
 5m zgrad r2d igrad r3d
 lo to 3 times td2
```

Setting precompiler options

Setting precompiler options

- Precompiler options allow parts of the pulse program to be used or to be neglected, depending on precompiler options
- This allows to have one pulse program, which can be used for solvent suppression either by presaturation or by WATERGATE, e.g.
- Optioned can be
  - defined in the pulse program
    - disadvantage: pulse program has to be modified when the option is changed
  - Acquisition parameter ZGOPTNS
    - several option flags can be set together

Setting precompiler options

Example: setting precompiler options within the pulse program:

presaturation or WATERGATE?

```
define PRESAT
;undef PRESAT
;define WATERGATE
undef WATERGATE

#ifdef PRESAT
  d12 pl9:f1
  d1 cw:f1
  d13 do:f1
  d12 pl1:f1
#endif PRESAT
#ifdef WATERGATE
  d1 pl1:f1
#endif WATERGATE
……..
```

Setting precompiler options

Example: setting precompiler options with ZGOPTNS:

presaturation or WATERGATE?

A)	**ZGOPTNS**	**-DPRESAT**	**used presaturation**
B)	**ZGOPTNS**	**-DWATERGATE**	**used WATERGATE**
C)	**ZGOPTNS**	**-DPRESAT -DWATERGATE**	**used presaturation and WATERGATE**

```
#ifdef PRESAT
  d12 pl9:f1
  d1 cw:f1
  d13 do:f1
  d12 pl1:f1
#endif PRESAT
#ifdef WATERGATE
  d1 pl1:f1
#endif WATERGATE
……..
```

The mc macro statement

(available since XWIN-NMR 3.1)

Pulse Programming: example HNCO

```
p16:gp4*EA                                  echo-antiecho gradient
 go=2 ph31 cpd3:f3
  d11 do:f3 wr #0 if #0 zd
  6m ip6*2 igrad EA                         increment phase, invert echo-antiech gradient
if "d30 < d21-p16-d16-4u" goto 98 conditional loop with runtime check

  d11 id10
  d11 dd30
  d11
  goto 99

98 d11 id10
   d11 dd30
   d11 id28

99 d12
  lo to 4 times l13
  d11 rd10 ip4
  d11 rd30
  d11 rd28
  lo to 5 times 2
  d11 id0
  lo to 6 times l3
  exit
```

Pulse Programming: the mc macro statement

Advantages of the mc statement:

simplified pulse programming:

One pulse program can be used for different 2D phase modes (TPPI, States or States-TPPI)

Pulse Programming: the mc macro statement

mc clause	possible mode	values of F1-FnMODE
F1QF	phase insensitive	QF
F1PH	phase sensitive	QSEQ, States, TPPI, States-TPPI
F1EA	Echo-Antiecho	Echo-Antiecho

The acquisition and processing parameter FnMODE define the mode for F1 (and F2, for 3D-experiments) quadrature

For details, see XWINNMR help -> Other topics -> Writing pulse program

Pulse Programming: the mc macro statement

up to XWINNMR 2.6

```
;noesytp

#include <Avance.incl>

"d0=3u"


1 ze
2 d1
3 p1 ph1
  d0
  p1 ph2
  d8
  p1 ph3
  go=2 ph31
  d1 wr #0 if #0 ip1 id0 zd
  lo to 3 times td1
exit
```

new: XWINNMR 3.0

```
;noesyph

#include <Avance.incl>

"d0=3u"


1 ze
2 d1
3 p1 ph1
  d0
  p1 ph2
  d8
  p1 ph3
  go=2 ph31
  d1 mc #0 to 2 F1PH(ip1, id0)
exit
```

Pulse Programming: the mc macro statement

Examples:

MC clause	t1 quadrature mode	action after	
		odd increment	even increment
F1PH(ip1, id0)	TPPI	ip1+ id0	ip1 + id0 again
F1PH(ip1, id0)	States-TPPI	ip1	id0
F1PH(rd10 & rd30 & ip4, id0)	TPPI	rd10+rd30+ip4+id0	
F1PH(rd10 & rd30 & ip4, id0)	States-TPPI	rd10+rd30+ip4	id0

Pulse Programming: the mc macro statement

Details

within XWINNMR, nD-pulse sequences will be expanded:

example for NOESY, F1-FnMODE = States-TPPI

```
define loopcounter ST1CNT
"ST1CNT = td1 / ( 2 ) “                          loop counter
"MCWRK = 0.500000 * d1“                          delays to ensure steady-state
"MCRES  = d1- d1“
        1 ze
        2                MCWRK
        LBLSTS1,         MCWRK          jump address
        LBLF1,           MCREST
        3 p1 ph1
        d0
        p1 ph2
        d8
        p1 ph3
        go=2 ph31
                MCWRK wr #0 if #0 zd ip1
                lo to LBLSTS1 times 2
                MCWRK id0
                lo to LBLF1 times ST1CNT
```

Pulse Programming: the mc macro statement

Example 1: **zg**

```
1 ze
2 30m
  d1
  p1 ph1
  go=2 ph31
  30m mc #0 to 2 F0(zd)
exit
```

New jump address added here

Jump address 2: just a delay

Pulse Programming: the mc macro statement

Example 2: **zgdc**

```
1 ze
   d11 pl12:f2
2 30m do:f2
   d11 cpd2:f2
   d1
   p1 ph1
   go=2 ph31
   30m do:f2 mc #0 to 2 F0(zd)
exit
```

New jump address added here

Jump address 2: switch OFF the decoupler!!!

go=2 loop goes to jump adress 2

Pulse Programming: the mc macro statement

Example 3: **zgpr**

```
1 ze
2 30m
  d12 pl9:f1
  d1 cw:f1 ph29
  4u do:f1
  d12 pl1:f1
  p1 ph1
  go=2 ph31
  30m mc #0 to 2 F0(zd)
exit
```

New jump address added here

Jump address 2: just a delay
power level setting done directly after

Pulse Programming: the mc macro statement

Example 4: **hsqcetgp**

```
1 ze
  d11 pl12:f2
2 d1 do:f2
3 (p1 ph1)
  d4 pl2:f2
  ……
go=2 ph31 cpd2:f2
  d1 do:f2 mc #0 to 2
     F1EA(igrad EA, id0 & ip3*2 & ip6*2 & ip31*2)
exit
```

New jump address added here

Jump address 2: switch OFF the decoupler!!!

Pulse Programming: the mc macro statement

Example 5: **hncogp3d**

```
1 d11 ze
   d11 pl16:f3
2 d11 do:f3
3 d1 pl1:f1
   p1 ph1
   d26 pl3:f3
   ……
  go=2 ph31 cpd3:f3
  d11 do:f3 mc #0 to 2
     F1PH(rd10 & rd29 & rd30 & ip4, id0)
     F2EA(igrad EA & ip6*2, id10 & id29 & dd30)
exit
```

New jump address added here

Jump address 2: switch OFF the decoupler!!!